Molecular Structure from a Single NMR Supersequence

Ēriks Kupče† and Tim D. W. Claridge‡

† Bruker UK Ltd., Banner Lane, Coventry, CV4 9GH, UK

‡ Department of Chemistry, University of Oxford, Chemistry Research Laboratory, Mansfield Road, Oxford, OX1 3TA, UK

Supporting Information

The ASAP-COSY Experiment

The effect of the short spinlock on the polarization distribution in 1H spectra of pamoic acid is demonstrated in Fig. S1. The intensities of peaks in the conventional 1D ^{1}H spectrum are shown along with the T_1 relaxation times in Fig S1a. These intensities become significantly distorted immediately after the ZZ-HMBC and HSQC modules are applied (see Fig. S1b). For instance, the resonance at 8.15 ppm with the shortest T_1 relaxation time recovers faster and is significantly stronger as compared to other peaks in the same ^{1}H-^{1}H *J*-coupling network. Following a short spinlock period of just 40 ms the relative intensities are largely restored (see Fig. S1c) and do not change significantly with a longer mixing time of 60 ms. Therefore in this particular sample 40 ms mixing time is considered to be optimum. Longer mixing times may be required in spin systems with small *J*-couplings.

The artifact suppression effect of the short spinlock period is demonstrated in Fig. S2. The ASAP-COSY pulse sequence is shown in Fig. S2a. In this experiment the ^{1}H magnetization is recovering during the free induction decay (FID) period, t_2 of the previous scan and the subsequent recovery delay, d_1 of the new scan. Therefore, longer acquisition times help both the sensitivity and resolution particularly in the ASAP-COSY experiments. This is followed by a short spinlock period, *SL* designed to homogenize the distribution of the ^{1}H polarization prior to the ASAP-COSY experiment and to reduce the fast pulsing artifacts. The conventional COSY spectrum recorded with a short repetition period of d_1 = 0.2 sec is shown in Fig. S2b. The same pulse sequence except for the included 40 ms adiabatic WURST-2 mixing period produces a clean spectrum under the same fast pulsing regime (see Fig. S2c).

The *F1* traces extracted from the two spectra at *ca* 8.15 ppm (*F2*) are compared in Fig. S3. The fast pulsing artifacts (marked with '*' in the blue trace) that essentially ruin the conventional COSY spectrum are effectively suppressed in the ASAP-COSY spectrum (the red trace). Note also that the relative intensity of the diagonal peak shows a welcome reduction in amplitude in the ASAP version of the experiment. It should be noted that unlike in the hetero nuclear ASAP experiments, such as ASAP-HMQC, the overall sensitivity in the ASAP-COSY spectra is reduced. This reflects the fact that the polarization is merely redistributed within the *J*-coupling networks of the molecule, whereas in the hetero nuclear experiments the bulk magnetization is serving as a source ('bath') of polarization.



Figure S1. a) The conventional 1D 1H spectrum of the aromatics region of pamoic acid (see inset in Fig. 2b) in DMSO-d_6 with T_1 relaxation times shown above the peaks; b) the 1D peak intensities measured for the t_2'' period with the pulse sequence (e) with spinlock (*SL*) time set to 0 (zero) ms; c) the same spectrum recorded with 40 ms adiabatic WURST-2 spinlock and d) 60 ms adiabatic WURST-2 spinlock; e) the pulse sequence that was used to record spectra (b) – (d); the filled bars denote 90 degree pulses whereas the hollow rectangles on ^{1}H and ^{13}C (=X) channels denote 180 degree pulses, the dotted rectangle in the HSQC module is an optional 180 degree pulse to provide multiplicity edited HSQC spectra; all ^{13}C 180 degree pulses are adiabatic CA WURST-20 pulses; g0 are spoiler gradients of arbitrary intensities, g1 and g2 are coherence selection gradients applied with a ratio of 2:1; delays $\Delta = 0.25/^{1}J(CH)$, $\varepsilon = 2\Delta$, $\delta = 0.5/^{n}J(CH)$; all pulses are applied with phase *x* unless indicated otherwise; ϕ_1 = *x, -x*; ϕ_2 = *x, x, -x, -x;* receiver phases, *rec* = *rec'* = *x, -x, -x, x*; *rec"* = *x, -x*; the WURST-2 adiabatic spinlock, *SL* was applied using 1 ms long adiabatic pulses with adiabaticity factor $Q = 3$ (generated using the *Wavemaker* tool within Bruker's TOPSPIN).



Figure S2. a) The ASAP-COSY pulse sequence; the filled bars denote 90 degree pulses, all pulses are applied with phase x unless indicated otherwise; ϕ_1 = *x*, *-x*; *rec* = *x*, *-x*; the adiabatic spinlock, *SL* was applied using 1 ms long adiabatic WURST-2 pulses with adiabaticity factor Q = 3; g0 are spoiler gradients of arbitrary intensities, g1 and g2 are coherence selection gradients applied with a ratio of 1:1; b) the conventional 2D COSY spectrum of the aromatics region of pamoic acid (see inset in Fig. 2b of the main text) in DMSO-d_6 recorded with the recovery delay, d_1 = 0.2 s; c) the ASAP-COSY spectrum recorded with identical experimental conditions and the adiabatic WURST-2 mixing period, *SL* of 40 ms; both spectra were recorded with 256 ms acquisition period, t_2 and plotted with similar peak intensities;



Figure S3. The F1 traces taken at 8.15 ppm (F2) from the 2D spectra of Figure S2 comparing the extent of artifacts in the conventional 2D COSY spectrum (in black) and the ASAP-COSY spectrum (in red). Note also the welcome reduction of the diagonal peak intensity in the ASAP-COSY spectra.

The NOAH-4 BSCN Experiment

The NOAH-4 BSCN pulse sequence is constructed by appending the NOESY module to the NOAH-3 BSC supersequence as shown in Fig. S4. The H-C ZZ-HMBC (B) and H-C HSQC (S) blocks are equivalent to those shown in Figs. 2a and 3a (see the main text) and are not shown explicitly for clarity and brevity reasons. The t_1-evolution period is shared by the ASAP-COSY and ASAP-NOESY modules. To allow for recording phase sensitive 2D NOESY spectra the ASAP-COSY module in this experiment must be phase sensitive in F1. Hence, the appropriate refocusing period, ζ is introduced in the t_1-evolution period in order to compensate for evolution during the coherence encoding gradient, g3.



Figure S4. The NOAH-4 BSCN pulse sequence; the (B) and (S) modules are equivalent to those shown in Fig 1S; all ζ delays are set to the same duration, typically 1.2 ms to accommodate a gradient and the associated recovery period; the NOESY module shares the t_1'' evolution time with the preceding COSY module and its mixing period (mix) includes the acquisition time t_2" of the COSY module; *SL* represents a short spinlock of 40-60 ms duration, g0 are spoiler gradients of arbitrary intensities, the coherence selection gradient ratio g3:g4 = 1:1, the sign of the g4 gradients is inverted in alternate increments for echo-anti-echo encoding; all pulses are applied with phase x unless indicated otherwise, ϕ_1 = *x, -x;* ϕ_2 = *x, x, -x, -x,* receiver phases *rec´´ = x, -x; rec´´´ = x, -x, -x, x;*

CASE: use of CMC-se for structure elucidation

The 1D H-1 and C-13 spectra were recorded on the same instrumentation prior to the main NOAH experiments to obtain preliminary information about the optimal spectral windows and the sample consistency as is customary, although technically not absolutely necessary. This took 2 sec (1 scan) and 24 sec (8 scans) respectively for each sample. This preliminary data may also be provided for the CMCse analysis for automatic rejection of minor artifacts and small impurities. The NOAH data were separated prior to the CASE analysis using the ***splitx*** au-program that is available from the online Bruker User Library, The Resonance Exchange: https://www.bruker.com/service/information-communication/nmr-pulse-program-lib/bruker-user-library.html
The spectra were labeled according to their type (SPECTYP) as required by the CMCse. The peak picking and spectral analysis is done automatically once the project properties (molecular mass, formula and solvent) are set. The rest of the CMCse workflow follows the standard procedures described in the CMCse manuals and in literature (see refs. [30, 31] in the main text).

Details

Chemical formula: $C_{12}H_{20}O_6$

Mass [Da]: 260.28
Solvent: DMSO
Description:



Descriptors

InChI: InChI=1/C12H20O6/c1-11(2)15-7-6(5-13)14-10-9(8(7)16-11)17-12(3,4)18-10/h6-10,13H,5H2,1-4H3

InChIKey: POORJMIIHXHXAV-UHFFFAOYNA-N

SMILES: OCC1OC2OC(OC2C3OC(OC13)(C)C)(C)C

Project: C:\Users\eriks.kupce\Desktop\MRdata\CMCdata\pyranose_BSCN\pyranose_BSCN

Report file: C:\Users\eriks.kupce\Desktop\MRdata\CMCdata\pyranose_BSCN\pyranose_BSCN\pyranose_BSCN.pdf



^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C12	1.28		(C12)	H11
C11	1.28		(C11)	H12
C9	1.34		(C9)	H10
C10	1.45		(C10)	H9
C8'	3.41		(C8)	H8
C8	3.49		(C8)	H7
C7	3.69		(C7)	H6
C4	4.24		(C4)	H5
C6	4.31		(C6)	H4
C5	4.57		(C5)	H3
H2	4.67			H2
C3	5.44		(C3)	H1

^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	24.68	3	C12
	25.34	3	C11
	26.32	3	C10
	26.41	3	C9
	60.41	2	C8
	68.68	1	C7
	70.43	1	C6
	70.44	1	C5
	70.59	1	C4
	96.1	1	C3
	108.03	0	C2
	108.5	0	C1

^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C12	1.28		(C12)	H11
C11	1.28		(C11)	H12
C9	1.34		(C9)	H10
C10	1.45		(C10)	H9
C8'	3.41		(C8)	H8
C8	3.49		(C8)	H7
C7	3.69		(C7)	H6
C4	4.24		(C4)	H5
C6	4.31		(C6)	H4
C5	4.57		(C5)	H3
H2	4.67			H2
C3	5.44		(C3)	H1



^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	24.68	3	C12
	25.34	3	C11
	26.32	3	C10
	26.41	3	C9
	60.41	2	C8
	68.68	1	C7
	70.43	1	C6
	70.44	1	C5
	70.59	1	C4
	96.1	1	C3
	108.03	0	C2
	108.5	0	C1



Explained Correlations

HMBC COSY ROESY



Incorrect Correlations

COSY



Chemical Shift Correlation





Details

Chemical formula: $C_{19}H_{22}O_6$

Mass [Da]: 346.38
Solvent: Acetone
Description:



Descriptors

InChI: InChI=1/C19H22O6/c1-9-7-17-8-18(9,24)5-3-10(17)19-6-4-11(20)16(2,15(23)25-19)13(19)12(17)14(21)22/h4,6,10-13,20,24H,1,3,5,7-8H2,2H3,(H,21,22)/f/h21H

InChIKey: IXORZMNAPKEEDV-PKSOQXRJNA-N

SMILES: O=C(O)C1C2C3(OC(=O)C2(C)C(O)C=C3)C4CCC5(O)C(=C)CC14C5

Project: C:\Users\eriks.kupce\Desktop\MRdata\CMCdata\giberellic_BSC\giberellic_BSC

Report file: C:\Users\eriks.kupce\Desktop\MRdata\CMCdata\giberellic_BSC\giberellic_BSC\Gibberellic_Acid.pdf



^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C19	1.21		(C19)	H18
C18'	1.71		(C18)	H17
C17'	1.75		(C17)	H16
C15'	1.81		(C15)	H15
C18	1.85		(C18)	H14
C15	1.93		(C15)	H12
C12	1.98		(C12)	H11
C17	2.06		(C17)	H10
C16'	2.24		(C16)	H9
C16	2.35		(C16)	H8
C13	2.69		(C13)	H7
C11	3.23		(C11)	H6
C9	4.05		(C9)	H5
C6'	4.89		(C6)	H4
C6	5.22		(C6)	H3
C4	5.89		(C4)	H2
C5	6.38		(C5)	H1

^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	14.16	3	C19
	16.8	2	C18
	38.89	2	C17
	43.0	2	C16
	44.58	2	C15
	49.88	0	C14
	50.91	1	C13
	51.04	1	C12
	52.58	1	C11
	53.41	0	C10
	69.24	1	C9
	77.1	0	C8
	90.37	0	C7
	105.98	2	C6
	132.01	1	C5
	133.12	1	C4
	157.92	0	C3
	172.65	0	C2
	178.39	0	C1

^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C19	1.21		(C19)	H18
C18'	1.71		(C18)	H17
C17'	1.75		(C17)	H16
C15'	1.81		(C15)	H15
C18	1.85		(C18)	H14
C15	1.93		(C15)	H12
C12	1.98		(C12)	H11
C17	2.06		(C17)	H10
C16'	2.24		(C16)	H9
C16	2.35		(C16)	H8
C13	2.69		(C13)	H7
C11	3.23		(C11)	H6
C9	4.05		(C9)	H5
C6'	4.89		(C6)	H4
C6	5.22		(C6)	H3
C4	5.89		(C4)	H2
C5	6.38		(C5)	H1



^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	14.16	3	C19
	16.8	2	C18
	38.89	2	C17
	43.0	2	C16
	44.58	2	C15
	49.88	0	C14
	50.91	1	C13
	51.04	1	C12
	52.58	1	C11
	53.41	0	C10
	69.24	1	C9
	77.1	0	C8
	90.37	0	C7
	105.98	2	C6
	132.01	1	C5
	133.12	1	C4
	157.92	0	C3
	172.65	0	C2
	178.39	0	C1



Explained Correlations

HMBC COSY



Incorrect Correlations

HMBC



Chemical Shift Correlation





All pulse programmes

1) NOAH-2 BS

```
;NOAH-2 BS: HMBC + me-HSQC
;Topspin 3 version 2018/04/09
;WaveMaker supported version
;with adiabatic refocussing and optional mult. editing in HSQC
;
;Ref: E. Kupce and T. D. W. Claridge, Chem. Commun., 2018, DOI: 10.1039/c8cc03296c
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=0.5s/cnst2"
"d3=0.25s/cnst2"
"d6=0.5s/cnst13"
"d0=3u"
"in0=inf1/2"
"l0=td1/4"

"DELTA=d2-p16-10u"
"DELTA1=p2+d0*2"
"DELTA2=d6-p16-10u"
"DELTA3=p16+d16+p2/2+4u+d0-p3*2/3.14159"
"DELTA6=d3-p14/2"
"DELTA7=d3+p14/2"

"DELTA8=p16+d16+p1+4u-p4/3.14159"
"DELTA9=d3-p14/2"
"DELTA10=d3+p14/2"
"DELTA12=DELTA7-p16-p3-de-p2/3.14-8u"

"acqt0=0"
baseopt_echo

1 ze
   30m pl12:f2
```

```
2 30m
3 3m do:f2
4 50u UNBLKGRAD
   4u pl1:f1
   p16:gp0
   4u
   (p1 ph1)
   4u
   p16:gp0*1.37
   50u BLKGRAD
   d1 st0
   p1 ph1                    ;ZZ-HMBC
   DELTA6
   (p14:sp3 ph4):f2
   (p2 ph1):f1
   DELTA7
   p1 ph1
   DELTA6 UNBLKGRAD
   (p14:sp3 ph4):f2
   (p2 ph1):f1
   DELTA7 pl2:f2 ; -HC-Hz
   p1 ph1              ; +HC(z) -Hx
   10u
   p16:gp0
   DELTA                ; J-filter
   (p3 ph1):f2
   10u
   p16:gp0*-1
   DELTA2
   (p3 ph5):f2
   DELTA3
   (p14:sp3 ph4):f2
   4u
   p16:gp1*EA*0.77
   d16
   d0
   (p2 ph1)
   d0
   4u
   p16:gp1*EA*0.77
   d16
   (p14:sp3 ph4):f2
   DELTA3 pl2:f2
   (p3 ph3):f2
   (p2 ph1)
   4u
   p16:gp2*0.77
   4u
```

```
   d16 ; BLKGRAD
   goscnp ph30             ;acquire HMBC

   4u pl2:f2
   (p3 ph1):f2
   50u ; UNBLKGRAD
   p16:gp0*1.77
   2m st
   (p1 ph1)                                    ;HSQC
   DELTA9
   (p14:sp3 ph1):f2
   (p2 ph1):f1
   DELTA10 pl2:f2
   (p1 ph2):f1
   (p3 ph3):f2
   DELTA8
   (p14:sp3 ph1):f2
   4u
   p16:gp1*EA
   d16
   d0
   (p2 ph1)
   d0
   4u
   p16:gp1*EA
   d16
   (p14:sp3 ph1):f2

#ifdef EDIT                ;optional HSQC multiplicity editing
      DELTA8
      (p2 ph1):f1
      (p14:sp3 ph6):f2
      d2
      (p2 ph1):f1
      (p14:sp3 ph6):f2
      d2 pl2:f2
#else
      DELTA8 pl2:f2
#endif

   (p3 ph5):f2
   (p1 ph1):f1
   DELTA9
   (p14:sp3 ph1):f2
   (p2 ph1):f1
   p16:gp2
   DELTA12 ; BLKGRAD
   4u pl2:f2
```

```
   (p3 ph1):f2
   4u pl12:f2
   go=2 ph30 cpd2:f2       ; acquire HSQC
   30m wr #0 if #0 zd igrad EA
   4u do:f2

   lo to 3 times 2
   1m id0
   1m ip3*2
   1m ip30*2
   lo to 4 times l0

50u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0
ph30=0 2 2 0



;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -     90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -     90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;sp3: f2 channel - shaped pulse 180 degree
;p16: homospoil/gradient pulse                                [1 msec]
;d0 : incremented delay (2D)                                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = nJ(XH) long range
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: 16
;NBL=2: number of NOAH modules
;FnMODE: echo-antiecho
```

```
; ~~~~~~~ use 'wvm -a' command to create these shapes ~~~~~~~~~~~~~~~~~~~~
;sp3:wvm:wu180C13: cawurst-20(260 ppm, 0.5 ms)
;cpd2:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms)

;use gradient ratio:      gp 1 : gp 2
;                          80 : 40.2       for C-13
;                          80 : 16.2       for N-15

;for z-only gradients:
;gpz0: 17.13%
;gpz1: 80%
;gpz2: 40.2% for C-13, 16.2% for N-15


;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100


;EDIT: for C-13 multiplicity editing start experiment with
;                  option -DEDIT (eda: ZGOPTNS)

;Processing
;use splitx au-program to split the data
;define user processing au programs (USERP1...P2) for automatic processing of all data sets
```

2) NOAH-3 BSC

```
;NOAH-3 BSC: HMBC + me-HSQC + COSY
;Topspin 3 version 2018/04/09
;WaveMaker supported version
;with adiabatic refocussing and optional mult. editing in HSQC
;
;Ref: E. Kupce and T. D. W. Claridge, Chem. Commun., 2018, DOI: 10.1039/c8cc03296c
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
```

```
"p2=p1*2"
"p4=p3*2"
"d2=0.5s/cnst2"
"d3=0.25s/cnst2"
"d6=0.5s/cnst13"
"d0=3u"
"d10=3u"
"in0=inf1/2"
"in10=inf1*cnst10"    ; cnst10 is a scaling factor swC/swH for COSY
"l0=td1/6"
"l9=(d9/(p45*20))"

"DELTA=d2-p16-10u"
"DELTA1=p2+d0*2"
"DELTA2=d6-p16-10u"
"DELTA3=p16+d16+p2/2+4u+d0-p3*2/3.14159"
"DELTA6=d3-p14/2"
"DELTA7=d3+p14/2"

"DELTA8=p16+d16+p1+4u-p4/3.14159"
"DELTA9=d3-p14/2"
"DELTA10=d3+p14/2"
"DELTA12=DELTA7-p16-p3-de-p2/3.14-8u"

"acqt0=0"
baseopt_echo

1 ze
   30m pl12:f2
2 30m
3 4m do:f2
4 50u UNBLKGRAD
   4u pl1:f1
   p16:gp0
   4u
   (p1 ph1)
   4u
   p16:gp0*1.37
   50u BLKGRAD
   d1 st0
   p1 ph1                    ;ZZ-HMBC
   DELTA6
   (p14:sp3 ph4):f2
   (p2 ph1):f1
   DELTA7
   p1 ph1
   DELTA6 UNBLKGRAD
   (p14:sp3 ph4):f2
```

```
(p2 ph1):f1
DELTA7 pl2:f2 ; -HC-Hz
p1 ph1              ; +HC(z) -Hx
10u
p16:gp0
DELTA                ; J-filter
(p3 ph1):f2
10u
p16:gp0*-1
DELTA2
(p3 ph5):f2
DELTA3
(p14:sp3 ph4):f2
4u
p16:gp1*EA*0.77
d16
d0
(p2 ph1)
d0
4u
p16:gp1*EA*0.77
d16
(p14:sp3 ph4):f2
DELTA3 pl2:f2
(p3 ph3):f2
(p2 ph1)
4u
p16:gp2*0.77
4u
d16 ; BLKGRAD
goscnp ph30          ;acquire HMBC

4u pl2:f2
(p3 ph1):f2
50u ; UNBLKGRAD
p16:gp0*1.77
5m st
(p1 ph1)                                    ;HSQC
DELTA9
(p14:sp3 ph1):f2
(p2 ph1):f1
DELTA10 pl2:f2
(p1 ph2):f1
(p3 ph3):f2
DELTA8
(p14:sp3 ph1):f2
4u
p16:gp1*EA
```

```
   d16
   d0
   (p2 ph1)
   d0
   4u
   p16:gp1*EA
   d16
   (p14:sp3 ph1):f2

#ifdef EDIT                    ;optional HSQC multiplicity editing
       DELTA8
       (p2 ph1):f1
       (p14:sp3 ph6):f2
       d2
       (p2 ph1):f1
       (p14:sp3 ph6):f2
       d2 pl2:f2
#else
       DELTA8 pl2:f2
#endif

   (p3 ph5):f2
   (p1 ph1):f1
   DELTA9
   (p14:sp3 ph1):f2
   (p2 ph1):f1
   p16:gp2
   DELTA12 ; BLKGRAD
   4u pl2:f2
   (p3 ph1):f2
   4u pl12:f2
   goscnp ph30 cpd2:f2      ;acquire HSQC
   50u do:f2

if "d9 > 1m"      ;ASAP-COSY
{
   50u ; UNBLKGRAD
   p16:gp0*1.77
   d16
   50u ; BLKGRAD

9 (p45:sp45 ph10):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph11):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph12):f1
```

```
   (p45:sp45 ph11):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph13):f1
   (p45:sp45 ph15):f1
   (p45:sp45 ph14):f1
   (p45:sp45 ph15):f1
   (p45:sp45 ph13):f1
   (p45:sp45 ph13):f1
   (p45:sp45 ph15):f1
   (p45:sp45 ph14):f1
   (p45:sp45 ph15):f1
   (p45:sp45 ph13):f1
   lo to 9 times l9
   4u pl1:f1       ;end mixing
}

   50u ; UNBLKGRAD
   p16:gp0
   5m st

   (p1 ph7)
   4u
   p16:gp3*EA
   d16
   d10                    ;COSY t1-evolution
   (p1 ph1)
   4u
   p16:gp3
   d16
   go=2 ph31          ;acquire H-H COSY
   30m wr #0 if #0 zd igrad EA

   lo to 3 times 2
   1m id0
   1m id10
   1m ip3*2
   1m ip30*2
   lo to 4 times l0

50u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
```

```
ph5=0 0 2 2
ph6=0
ph7=0 2
ph10=0
ph11=60
ph12=150
ph13=180
ph14=240
ph15=330
ph30=0 2 2 0
ph31=0 2


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;p45: f1 channel -  180 degree CAWURST-2 pulse
;sp3: f2 channel - shaped pulse 180 degree
;sp45: f1 channel - CAWURST-2 pulse (180 degree)
;p16: homospoil/gradient pulse                          [1 msec]
;d0 : incremented delay (2D)                            [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9: ASAP mixing duration [0 (no mixing) or 40-60 ms]
;d11: delay for disk I/O                                [5 msec]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = nJ(XH) long range
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns:  2 * n
;ds: 16
;NBL=3: number of NOAH modules
;FnMODE: echo-antiecho

; ~~~~~~~ use 'wvm -a' command to create these shapes ~~~~~~~~~~~~~~~~~~~
;sp3:wvm:wu180C13: cawurst-20(260 ppm, 0.5 ms)
;cpd2:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms)
;sp45:wvm:wuASAP: cawurst-2(30 ppm, 1.0 ms; Q=3)

;use gradient ratio:     gp 1 : gp 2 : gp 3
;                           80 : 40.2 : 25      for C-13
;                           80 : 16.2 : 25      for N-15
```

;for z-only gradients:
;gpz0: 17.13%
;gpz1: 80%
;gpz2: 40.2% for C-13, 16.2% for N-15
;gpz3: 25%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100

;EDIT: for C-13 multiplicity editing start experiment with
; option -DEDIT (eda: ZGOPTNS)

;Processing
;use splitx au-program to split the data
;define user processing au programs (USERP1...P3) for automatic processing of all data sets

3) NOAH-4 BSCN

;NOAH-4 BSCN: HMBC + me-HSQC + COSY + NOESY
;Topspin 3 version 2018/04/09
;WaveMaker supported version
;with adiabatic refocussing and optional mult. editing in HSQC
;
;Ref: E. Kupce and T. D. W. Claridge, Chem. Commun., 2018, DOI: 10.1039/c8cc03296c
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d2=0.5s/cnst2"
"d3=0.25s/cnst2"
"d6=0.5s/cnst13"
"d0=3u"
"d10=3u"
"in0=inf1/2"

```
"in10=inf1*cnst10"    ; cnst10 is a scaling factor swC/swH for COSY
"l0=td1/8"
"l9=(d9/(p45*20))"

"d22=p16+d16"
"DELTA=d2-p16-10u"
"DELTA1=p2+d0*2"
"DELTA2=d6-p16-10u"
"DELTA3=d22+p2/2+4u+d0-p3*2/3.14159"
"DELTA6=d3-p14/2"
"DELTA7=d3+p14/2"

"DELTA8=d22+p1+4u-p4/3.14159"
"DELTA9=d3-p14/2"
"DELTA10=d3+p14/2"
"DELTA12=DELTA7-p16-p3-de-p2/3.14-8u"

"DELTA4=d22+4u+d10"
"DELTA5=d8-d22-3m-aq-p3-p16-62u"    ; adjusts noe mixing time

"acqt0=0"
baseopt_echo

1 ze
   30m pl12:f2
2 30m
3 6m do:f2
4 50u UNBLKGRAD
   4u pl1:f1
   p16:gp0
   4u
   (p1 ph1)
   4u
   p16:gp0*1.37
   50u BLKGRAD
   d1 st0
   p1 ph1                    ;ZZ-HMBC
   DELTA6
   (p14:sp3 ph4):f2
   (p2 ph1):f1
   DELTA7
   p1 ph1
   DELTA6 UNBLKGRAD
   (p14:sp3 ph4):f2
   (p2 ph1):f1
   DELTA7 pl2:f2 ; -HC-Hz
   p1 ph1              ; +HC(z) -Hx
   10u
```

```
p16:gp0
DELTA                   ; J-filter
(p3 ph1):f2
10u
p16:gp0*-1
DELTA2
(p3 ph5):f2
DELTA3
(p14:sp3 ph4):f2
4u
p16:gp1*EA*0.77
d16
d0
(p2 ph1)
d0
4u
p16:gp1*EA*0.77
d16
(p14:sp3 ph4):f2
DELTA3 pl2:f2
(p3 ph3):f2
(p2 ph1)
4u
p16:gp2*0.77
4u
d16
goscnp ph30             ; acquire HMBC

4u pl2:f2
(p3 ph1):f2
50u
p16:gp0*1.77
5m st
(p1 ph1)                                ;HSQC
DELTA9
(p14:sp3 ph1):f2
(p2 ph1):f1
DELTA10 pl2:f2
(p1 ph2):f1
(p3 ph3):f2
DELTA8
(p14:sp3 ph1):f2
4u
p16:gp1*EA
d16
d0
(p2 ph1)
d0
```

```
   4u
   p16:gp1*EA
   d16
   (p14:sp3 ph1):f2

#ifdef EDIT                          ;optional HSQC multiplicity editing
      DELTA8
      (p2 ph1):f1
      (p14:sp3 ph6):f2
      d2
      (p2 ph1):f1
      (p14:sp3 ph6):f2
      d2 pl2:f2
#else
      DELTA8 pl2:f2
#endif

   (p3 ph5):f2
   (p1 ph1):f1
   DELTA9
   (p14:sp3 ph1):f2
   (p2 ph1):f1
   p16:gp2
   DELTA12 ; BLKGRAD
   4u pl2:f2
   (p3 ph1):f2
   4u pl12:f2
   goscnp ph30 cpd2:f2      ; acquire HSQC
   50u do:f2

if "d9 > 1m"      ;ASAP-COSY
{
   50u ; UNBLKGRAD
   p16:gp0*1.77
   d16
   50u ; BLKGRAD

9 (p45:sp45 ph10):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph11):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph11):f1
   (p45:sp45 ph12):f1
   (p45:sp45 ph10):f1
   (p45:sp45 ph13):f1
```

```
    (p45:sp45 ph15):f1
    (p45:sp45 ph14):f1
    (p45:sp45 ph15):f1
    (p45:sp45 ph13):f1
    (p45:sp45 ph13):f1
    (p45:sp45 ph15):f1
    (p45:sp45 ph14):f1
    (p45:sp45 ph15):f1
    (p45:sp45 ph13):f1
    lo to 9 times l9
    4u pl1:f1       ;end mixing
}

    50u ; UNBLKGRAD
    p16:gp0
    5m st

    (p1 ph28)
    DELTA4
    (p2 ph1)
    4u
    p16:gp3
    d16
    d10             ; t1-evolution
    (p1 ph1)
    4u
    p16:gp3*EA
    d16
    goscnp ph28     ;acquire H-H COSY
    4u
    (p3 ph1):f2
    4u
    50u
    p16:gp4
    DELTA5 st
    p1 ph5
    d22
    de
    4u
    (p2 ph5)
    4u
    p16:gp3*EA
    d16
    go=2 ph29       ; acquire phase sensitive H-H NOESY

    30m wr #0 if #0 zd igrad EA

    lo to 3 times 2
```

```
   1m id0
   1m id10
   1m ip3*2
   1m ip28*2
   1m ip29*2
   1m ip30*2
   lo to 4 times l0

50u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0
ph7=0 2
ph10=0
ph11=60
ph12=150
ph13=180
ph14=240
ph15=330
ph28=0 2
ph29=0 2 2 0
ph30=0 2 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -    90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -    90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;p45: f1 channel -    180 degree CAWURST-2 pulse
;sp3: f2 channel - shaped pulse 180 degree
;sp45: f1 channel - CAWURST-2 pulse (180 degree)
;p16: homospoil/gradient pulse                                  [1 msec]
;d0 : incremented delay (2D)                                    [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9: ASAP mixing duration [0 (no mixing) or 40-60 ms]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = nJ(XH) long range
```

```
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: 16
;NBL=4: number of NOAH modules
;FnMODE: echo-antiecho

; ~~~~~~~ use 'wvm -a' command to create these shapes ~~~~~~~~~~~~~~~~~~~
;sp3:wvm:wu180C13: cawurst-20(260 ppm, 0.5 ms)
;cpd2:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms)
;sp45:wvm:wuASAP: cawurst-2(30 ppm, 1.0 ms; Q=3)

;use gradient ratio:        gp 1 : gp 2 : gp 3 : gp 4
;                              80 : 40.2 : 25 : 25        for C-13
;                              80 : 16.2 : 25 : 25      for N-15

;for z-only gradients:
;gpz0: 17.13%
;gpz1: 80%
;gpz2: 40.2% for C-13, 16.2% for N-15
;gpz3: 25%
;gpz4: 25%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100

;EDIT: for C-13 multiplicity editing start experiment with
;                        option -DEDIT (eda: ZGOPTNS)

;Processing
;use splitx au-program to split the data
;define user processing au programs (USERP1...P4) for automatic processing of all data sets
```

4) ASAP-COSY

```
;ASAP-COSY
;Topspin 3 version 2018/04/09
;WaveMaker supported version
;with adiabatic mixing to equalize the partially recovered magnetization
;
;Ref: E. Kupce and T. D. W. Claridge, Chem. Commun., 2018, DOI: 10.1039/c8cc03296c
;
;$CLASS=HighRes
```

```
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d0=3u"
"in0=inf1"
"l9=(d9/(p45*20))"
"d11=30m"

1 ze
2 d11 pl1:f1
   d1

   if "d9 > 1m"      ;begin Adiabatic Mixing
   {
      50u UNBLKGRAD
      p16:gp0*1.77
      d16
      50u BLKGRAD

9     (p45:sp45 ph10):f1
      (p45:sp45 ph12):f1
      (p45:sp45 ph11):f1
      (p45:sp45 ph12):f1
      (p45:sp45 ph10):f1
      (p45:sp45 ph10):f1
      (p45:sp45 ph12):f1
      (p45:sp45 ph11):f1
      (p45:sp45 ph12):f1
      (p45:sp45 ph10):f1
      (p45:sp45 ph13):f1
      (p45:sp45 ph15):f1
      (p45:sp45 ph14):f1
      (p45:sp45 ph15):f1
      (p45:sp45 ph13):f1
      (p45:sp45 ph13):f1
      (p45:sp45 ph15):f1
      (p45:sp45 ph14):f1
      (p45:sp45 ph15):f1
      (p45:sp45 ph13):f1

      lo to 9 times l9
```

```
      4u pl1:f1               ;end mixing
   }

   50u UNBLKGRAD
   p16:gp0
   2m

   (p1 ph1)
   4u
   p16:gp1*EA
   d16
   d0
   (p1 ph3)
   4u
   p16:gp2
   d16
   4u BLKGRAD
   go=2 ph31
   d11 mc #0 to 2 F1EA(calgrad(EA), caldel(d0, +in0) & calph(ph1, +180) & calph(ph31, +180))
exit

ph1=0 2 2 0
ph3=0 2
ph10=0
ph11=60
ph12=150
ph13=180
ph14=240
ph15=330
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;p1 : f1 channel -    90 degree high power pulse
;p45: f1 channel -    180 degree CAWURST-2 pulse
;sp45: f1 channel - CAWURST-2 pulse (180 degree)
;d0 : incremented delay (2D)                                    [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d9: ASAP mixing duration [0 (no mixing) or 40-60 ms]
;d16: delay for homospoil/gradient recovery
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: 16
;td1: number of increments
;FnMODE: echo-antiecho

; ~~~~~~~ use 'wvm -a' command to create these shapes ~~~~~~~~~~~~~~~~~~~
```

;sp45:wvm:wuASAP: cawurst-2(30 ppm, 1.0 ms; Q=3)

;use gradient ratio: gp 1
; 80

;for z-only gradients:
;gpz1: 80%

;use gradient files:
;gpnam1: SMSQ10.100

;Processing
;use xfb and xf2m